

list,map,count,loop,Splat Operator

在上一节的代码上修改，大部分代码跟上节一样，这里就列出被更改的文件

1, variable.tf 就添加了 list 跟 map

```
# AWS EC2 Instance Type - List
variable "instance_type_list" {
  description = "EC2 Instance Type"
  type = list(string)
  default = ["t2.micro", "t2.small"]
}

# AWS EC2 Instance Type - Map
variable "instance_type_map" {
  description = "EC2 Instance Type"
  type = map(string)
  default = {
    "dev" = "t2.micro"
    "qa"  = "t2.small"
    "prod" = "t2.large"
  }
}
```

2, ec2-setting.tf 文件 做了一些修改

```
resource "aws_instance" "myec2vm" {
  ami = data.aws_ami.amzlinux2.id

  # instance_type = var.instance_type
  # instance_type = var.instance_type_list[0] //reference List value
  instance_type = var.instance_type_map["dev"] //reference map value

  user_data = file("${path.module}/nginx-install.sh")
  key_name = var.instance_keypair
  vpc_security_group_ids = [aws_security_group.vpc-ssh.id, aws_security_group.vpc-web.id]

  count = 2      # 一次性创建两个ec2实例

  tags = {
    "Name" = "EC2 Demo-${count.index}"    #count.index指的是count的0, 1两个ec2实例
  }
}
```

3, output-values.tf 改了很多

```

# Output - For Loop with List
output "for_output_list" {
  description = "For Loop with List"
  value = [for instance in aws_instance.myec2vm: instance.public_dns ]
}

# Output - For Loop with Map
output "for_output_map1" {
  description = "For Loop with Map"
  value = {for instance in aws_instance.myec2vm: instance.id => instance.public_dns}
}

# Output - For Loop with Map Advanced
output "for_output_map2" {
  description = "For Loop with Map - Advanced"
  value = {for c, instance in aws_instance.myec2vm: c => instance.public_dns}
}

# Output Latest Generalized Splat Operator - Returns the List
output "latest_splat_instance_publicdns" {
  description = "Generalized Splat Expression"
  value = aws_instance.myec2vm[*].public_dns
}

```

```

TERMINAL  PROBLEMS  DEBUG CONSOLE  JUPYTER  OUTPUT  zsh - list-map-count-loop-splat
Outputs:
for_output_list = [
  "ec2-35-77-48-70.ap-northeast-1.compute.amazonaws.com",
  "ec2-52-68-39-110.ap-northeast-1.compute.amazonaws.com",
]
for_output_map1 = {
  "i-0b745a603dddb47ef" = "ec2-52-68-39-110.ap-northeast-1.compute.amazonaws.com"
  "i-0e755ccb26edab5c8" = "ec2-35-77-48-70.ap-northeast-1.compute.amazonaws.com"
}
for_output_map2 = {
  "0" = "ec2-35-77-48-70.ap-northeast-1.compute.amazonaws.com"
  "1" = "ec2-52-68-39-110.ap-northeast-1.compute.amazonaws.com"
}
latest_splat_instance_publicdns = [
  "ec2-35-77-48-70.ap-northeast-1.compute.amazonaws.com",
  "ec2-52-68-39-110.ap-northeast-1.compute.amazonaws.com",
]

```

4, list,map,count,loop,Splat Operator文件夹里执行命令

```

terraform init      // 自动根据terraform-setting.tf 生成 [.terraform]跟[.terraform.lock.hcl] 两个文件

terraform validate // 验证配置是否正确

terraform plan      // 列出将要做的事, 配置是什么

terraform apply -auto-approve

// 执行这条命令后, ec2实例建立、控制台会打印出下图的内容, output-values.tf 文件定义的内容

```

5,终结ec2实例,aws管理画面的实例跟security groups会被删除

```
# Terraform Destroy
terraform destroy -auto-approve

# Clean-Up Files
rm -rf .terraform*
rm -rf terraform.tfstate*
```

github link

<https://github.com/no744634936/terraform/tree/main/3-list-map-count-loop-splat>